

```

RegKey = RegistryOpenKey(CurrentUser!; "Software\ABBYY\Temp\Export\Word Perfect")
FileNameToOpen:=RegistryQueryValue(RegKey; "ExportFileName")

MessageTitle="Sending from FineReader"
Assign(hWnd; AppLocate("*FineReader*"))
IF(hWnd)
    SetDefaultParent(hWnd)
ENDIF
IF(StrLen(FileNameToOpen)<>0)
    FileExists(IsFileExist; FileNameToOpen)
    IF(IsFileExist)
        Display (Off!)
        ONERROR(NoNewFile)
        FileNew()
        ONERROR(NormalEnd)

        FileInsert(Filename:FileNameToOpen; AutoDetect:Yes!; InsertIntoDoc:Insert!)
        FileDelete(Filename:FileNameToOpen)
        Display (On!)

        Assign(hWnd; AppLocate("*WordPerfect*"))
        IF(hWnd)
            AppActivate(hWnd)
        ENDIF
    ELSE
        MessageBox(retValue; MessageTitle; "There is no file for export."; TaskModal! |
            IconAsterisk!)
    ENDIF
ELSE
    MessageBox(retValue; MessageTitle; "ExportFileName is empty."; TaskModal! |
        IconAsterisk!)
ENDIF

Label(NormalEnd)
Quit

Label(NoNewFile)
Assign(hWnd; AppLocate("*WordPerfect*"))
IF(hWnd)
    AppActivate(hWnd)
    SetDefaultParent(hWnd)
ENDIF
MessageBox(retValue;"WordPerfect Error"; "Can't create a new document. Close one or more
documents and try again."; TaskModal! | IconAsterisk!)
Quit

```